
Anonyma funktioner igen
En funktion som inte är namngiven kallas för en anonym funktion.

fn x => x+1

formellt 
argument

funktions-
kropp

När man anropar funktionen så beräknas funktionskroppen som om
x bytts ut mot argumentet i anropet. x är bunden i funktionen.

x är en godtycklig identifierare.

Funktionens typ bestäms av typen hos det formella argumentet och 
funktionskroppen. Funktionen ovan har typen int -> int.

(fn x => x+1)(4*2) -––> (fn x => x+1) 8 -––> 8+1 -––> 9

Funktioner är ”första klassens objekt” i ML
Kom ihåg att funktioner kan behandlas som datastrukturer i ML.
• Funktioner kan vara argument till andra funktioner
• Värdet av en funktion kan också vara en funktion
• Funktioner kan förekomma som delar av datastrukturer
  (t.ex. element i listor).

Funktioner kan vara argument
fun dotwice(f,x) = f(f x);
fun square x = x*x;

dotwice(square,3) -––> square(square 3)
-––> square(3*3) -––> square 9 -––> 9*9 -––> 81

dotwice(fn x=>x+1,3) -––> (fn x=>x+1)((fn x=>x+1) 3)
-––> (fn x=>x+1)(3+1) -––> (fn x=>x+1) 4 -––> 4+1 -––> 5

Vilken är typen av dotwice?
(Prova gärna genom att skriva in definitionen i ML.) 

En första ordningens funktion har argument som inte är funktioner.
En högre ordningens funktion har något funktionsargument.

merge sort igen
7 1 5 7 3 2 12 8

7 5 3 12 1 7 2 8

split

3 5 7 12 1 2 7 8

mergesort mergesort

1 2 3 5 7 7 8 12

merge

Varför är inte merge sort polymorf
- fun sort [] = []
=   | sort [x] = [x]
=   | sort xs =
=       let
=         val (ys, zs) = split xs
=       in
=         merge(sort ys, sort zs)
=     end;
val sort = fn : int list -> int list
 
Algoritmen för merge sort bryr sig inte om vilken typ det är på de 
data man sorterar. Så varför är inte funktionen sort polymorf?

Original-merge
3 5 7 12 1 2 7 8

1 2 3 5 7 7 8 12

merge

(* merge(L,M)
   TYPE: int list*int list -> int list
   PRE: L och M är sorterade i stigande ordning.
   POST: En lista med alla element i L och M i stigande 
ordning.
   EX.: merge([3,5,7,12],[1,2,7,8]) = [1,2,3,5,7,7,8,12] *)
(* VARIANT: (length l1)*(length l2) *)
fun merge ([], M) = M
  | merge (L, []) = L
  | merge (L as x::xs, M as y::ys) =
      if x > y then
        y :: merge (L, ys)
      else
        x :: merge (xs, M);

Polymorf merge
(* merge(less,L,M)
   TYPE: ('a*'a->bool)*'a list*'a list -> 'a list
   PRE: less är en ordningsrelation.
        L och M är sorterade i stigande ordning
        enligt less.
   POST: En lista med alla element i L och M i 
         stigande ordning enligt relationen less.
   EX.: merge(op <,[3,5,7,12],[1,2,7,8])
                           = [1,2,3,5,7,7,8,12] *)
(* Variant: (length L)*(length M) *)
fun merge(_, [], M) = M
  | merge(_, L, []) = L
  | merge(less, L as x::xs, M as y::ys) =
      if less(y,x) then
        y :: merge(less,L,ys)
      else
        x :: merge(less,xs,M);

PKD 2010/11 moment 9 Sida 8 Uppdaterad 2010-12-02



Notationen op
Funktioner som är infixa operatorer kan inte direkt vara argument:
  - merge(<,[3,5,7,12],[1,2,7,8]);
  ! Toplevel input:
  ! merge(<,[3,5,7,12],[1,2,7,8]);
  !       ^
  ! Ill-formed infix expression

ML förväntar sig argument före och efter <. Lösning: skriv op före <.
op ”stänger av” den speciella syntaktiska betydelsen hos <.

  - merge(op <,[3,5,7,12],[1,2,7,8]);
  > val it = [1, 2, 3, 5, 7, 7, 8, 12] : int list

Varning: foo(op *) gör inte vad man tror. Åtgärd: foo(op * )

Varför?                                                                    blank

Polymorf merge sort
(* polysort(less,L)
   TYPE: ('a*'a->bool)*'a list -> 'a list
   PRE: less är en ordningsrelation
   POST: En lista med alla element i L i stigande
         ordning enligt less.
   EXAMPLE: polysort(op <,[5,7,3,12,1,7,2,8,13])
                      = [1,2,3,5,7,7,8,12,13] *)
(* VARIANT: length L *)
fun polysort(_,[]) = []
  | polysort(_,[x]) = [x]
  | polysort(less,xs) =
     let
       val (ys, zs) = split xs
     in
       merge(less,polysort(less,ys), 
                  polysort(less, zs))
     end;

Mer komplicerade relationer
Sortera en lista av namn uttryckta som efternamn-förnamn-par:
val namnlista =
[("Bergkvist","Erik"),("Andersson","Arne"),
 ("Björklund","Henrik"),("Eriksson","Lars-Henrik"),
 ("Andersson","Petra"),("Berg","Stina")];

När kommer ett namn före ett annat?
Efternamnet skall jämföras först!
fun lessName((e1:string,f1:string),(e2,f2)) =
      e1 < e2 orelse e1 = e2 andalso f1 < f2;

Varför behövs typdeklarationerna (:string)?
polysort(lessName,namnlista) =
[("Andersson","Arne"),("Andersson","Petra"),
 ("Berg","Stina"),("Bergkvist","Erik"),
 ("Björklund","Henrik"),("Eriksson","Lars-Henrik")]

Värdet av en funktion kan vara en funktion
fun selectOpr s =
  case s of
      "plus" => op +
    | "minus" => op -
    | "times" => op *
    | "div" => op div
    | "sumto" => sumRange
    | "mean" => fn (x,y) => (x+y) div 2;
(sumRange(i,n) summerar heltalen från i till n.)
selectOpr "plus" -––> op +
selectOpr "mean" -––> fn (x,y) => (x+y) div 2
(selectOpr "plus") (4,6) -––> (op +)(4,6) -––> 10
(selectOpr "sumto") (4,6) -––> sumRange(4,6) -––> 15

Vilken är typen hos selectOpr?

Diverse om användning av funktionsvärden
Funktionsapplikation i ML är vänsterassociativ. Ett uttryck som
(f x) y kan lika gärna skrivas  f x y .
ML kan inte skriva ut funktionsvärden. Funktionsvärden ersätts med 
symbolen fn.
- selectOpr "plus";
> val it = fn : int * int -> int

Typen för selectOpr är  string->(int*int->int).
Typkonstruktorn -> är högerassociativ så vi kan i detta fall ta bort 
parenteserna och skriva:  string->int*int->int.

Lite annorlunda syntaxträd
Operatorn lagras i ett fält i varje nod.

–

+

3
2

*
x

13+2*x-1  :

                 
datatype expr = Const of int
              | Var of string
              | Op of string*expr*expr;

Op("minus",Op("plus",Const 3,
                     Op("times",Const 2,Var "x")),
           Const 1)

Beräkning av uttryck 
selectOpr används för att välja funktion givet informationen om 
operator i trädet.

(* eval(expr,table)
   TYPE: expr*(string*int) list->int
   PRE: Alla variabler i expr finns i table.
        Ingen division med 0 förekommer i expr.
   POST: Värdet av expr uträknat.
   EX: eval(Op("minus",
               Op("plus",Const 3,
                         Op("times",Const 2,Var "x")),
                  Const 1),
            insert([],"x",4)) = 10 *)
(* VARIANT: Storleken hos expr *)
fun eval(Const c, _) = c
  | eval(Var v, values) = search(values,v)
  | eval(Op(opr, e1, e2),values) =
      selectOpr opr (eval(e1,values),eval(e2,values));

Tabell med funktionsvärden 
I stället för att använda selectOpr för att välja funktion kan vi 
skapa en tabell som kopplar operatorerna i trädet till ML-funktioner.

val optable =
  [("plus", op +),
   ("minus", op -),
   ("times", op * ),
   ("div", op div),
   ("sumto", sumRange),
   ("mean", fn (x,y) => (x+y) div 2)];

Vilken typ har optable?

Beräkning av uttryck med funktionstabell 
search används för att hitta avsedd funktion givet informationen 
om operator i trädet.
(* eval(expr,table)
   TYPE: expr*(string*int) list->int
   PRE: Alla variabler i expr finns i table.
        Ingen division med 0 förekommer i expr.
   POST: Värdet av expr uträknat.
   EX: eval(Op("minus",
               Op("plus",Const 3,
                         Op("times",Const 2,Var "x")),
                  Const 1),
            insert([],"x",4)) = 10 *)
(* VARIANT: Storleken hos expr *)
fun eval(Const c, _) = c
  | eval(Var v, values) = search(values,v)
  | eval(Op(opr, e1, e2),values) =
      search(optable,opr) 
            (eval(e1,values),eval(e2,values));

Syntaxträd med funktioner i noderna
Själva ML-funktionen som skall utföra beräkningen lagras i ett fält i 
varje nod.

–

+

3
2

*
x

13+2*x-1  :

                 
datatype expr = Const of int
              | Var of string
              | Op of (int*int->int)*expr*expr;

Op(op -,Op(op +,Const 3,
                Op(op *,Const 2,Var "x")),
           Const 1)

Beräkning med funktioner i noderna
Nu behövs ingen översättning utan ML-funktionen kan anropas 
direkt!

(* eval(expr,table)
   TYPE: expr*(string*int) list->int
   PRE: Alla variabler i expr finns i table.
        Ingen division med 0 förekommer i expr.
   POST: Värdet av expr uträknat.
   EX: eval(Op(op -,
               Op(op +,Const 3,
                       Op(op *,Const 2,Var "x")),
                  Const 1),
            insert([],"x",4)) = 10 *)
(* VARIANT: Storleken hos expr *)
fun eval(Const c, _) = c
  | eval(Var v, values) = search(values,v)
  | eval(Op(opr, e1, e2),values) =
      opr(eval(e1,values),eval(e2,values));

Mönster vid listrekursion
fun squareList [] = []
  | squareList (first::rest) =
      square first::squareList rest;
fun negList [] = []
  | negList (first::rest) = ~first::negList rest;
Dessa funktioner har samma struktur – de skiljer sig bara i 
funktionsnamnen och de rödmarkerade delarna.
Man kan skriva en högre ordningens funktion mapl som fångar 
strukturen –– abstraherar bort skillnaderna –– och som har ett 
funktionsargument för skillnaden:
fun mapl(f,[]) = []
  | mapl(f,first::rest) = f first::mapl(f,rest);
fun squareList l = mapl(square,l);
fun negList l = mapl(~,l);

Mönster vid listrekursion (forts.)
fun mapl(f,[]) = []
  | mapl(f,first::rest) = f first::mapl(f,rest);
fun squareList l = mapl(square,l);

squareList [1,~2,4]
-––> mapl(square,[1,~2,4])
-––> square 1::mapl(square,[~2,4])
-––> 1*1::mapl(square,[~2,4])
-––> 1::mapl(square,[~2,4])
-––> 1::square ~2::mapl(square,[4])
-––> 1:: ~2* ~2::mapl(square,[4])
-––> 1::4::mapl(square,[4])
-––> 1::4::square 4::mapl(square,[])
-––> 1::4::4*4::mapl(square,[])
-––> 1::4::16::mapl(square,[])
-––> 1::4::16::[]
= [1,4,16]

”Stuvade” funktioner
fun plus(x,y) = x+y;        (plus:  int*int->int)
fun plus' x = fn y => x+y;  (plus':  int->int->int)

plus(4,6) –––> 4+6 –––> 10
plus' 4 6 –––> (fn y => 4+y)6 –––> 4+6 –––> 10
• plus' är en stuvad (curried – efter H.B. Curry) form av plus.
• plus' tar ”ett argument i taget”.
• Funktionen som är värdet av plus' ”minns” bindningen av x.
val add1 = plus' 1;    (add1 nu bunden till fn y => 1+y)
add1 3 –––> (fn y => 1+y)3 –––> 1+3 –––> 4
fun har en speciell form för att definiera stuvade funktioner:
    fun plus' x y = x+y;
är samma definition av plus' som den ovan.

Beräkning av funktioner
En stuvad form av mapl finns inbyggd i ML under namnet map.

mapl(f,l) = map f l

fun squareList l = map square l;

map kan också användas för att beräkna funktionen squareList:

val squareList = map square;

Beräkning av specialiserade 
sorteringsfunktioner

Antag att vi skrev den polymorfa polysort som en stuvad funktion 
med typ ('a*'a->bool) -> 'a list -> 'a list
Då går det lätt att göra specialversioner för olika typer:
- val sortInt = polysort op <;
val sortInt = fn : int list -> int list
- sortInt [5,7,3,12,1,7,2,8,13];
val it = [1,2,3,5,7,7,8,12,13] : int list
- val sortName = polysort lessName;
val sortName = fn : (string * string) list -> 
(string * string) list
- sortName namnlista;
val it = 
[("Andersson","Arne"),("Andersson","Petra"),("Berg
","Stina"),……]: (string * string) list

Mer komplicerade programmönster
fun length [] = 0
  | length (first::rest) = 1+length rest;
fun addList [] = 0
  | addList (first::rest) = first+addList rest;
fun append([],y) = y
  | append(first::rest,y) = first::append(rest,y);

Gemensamt för dessa:
• En operation utförs på första elementet av listan och resultatet
  av det rekursiva anropet (blått).
• Det finns ett bestämt värde (”startvärde”) vid basfallet (rött).
Att length passar in i mönstret syns bättre om man skriver så här:
  | length (first::rest) =
             (fn (x,y) => 1+y)(first,length rest);

Funktionen foldr
Den inbyggda funktionen foldr utför mer generell rekursion över 
listor än map. 
foldr kan utläsas som ”fold from right”. 
(* foldr f e l
   TYPE: ('a*'b->'b) -> 'b -> 'a list -> 'b
   PRE: (ingen)
   POST: f(x1,f(x2,...f(xn,e)...)) om l=[x1,x2,...,xn]
*)
(* VARIANT: längden av l *)
fun foldr f e [] = e
  | foldr f e (first::rest) =
      f(first,foldr f e rest);

Beräkning av funktioner med foldr
  fun foldr f e [] = e
    | foldr f e (first::rest) =
        f(first,foldr f e rest);
Vad är värdet av foldr (op +) 0? f binds till (op +) och e 
binds till 0. Resultatet blir alltså en funktion som är definierad som: 
  fun foo [] = 0
    | foo (first::rest) =
        (op +)(first,foo rest);
...vilket (utom syntaxen) är den rekursiva definitionen av addList. 
(Den beräknade funktionen är anonym – jag har kallat den för foo.)
addList kan alltså definieras som:
  val addList = foldr (op +) 0;
eller – kanske lite tydligare –
  fun addList l = foldr (op +) 0 l;

Användning av foldr
Exempel på användning av foldr:
fun length l = foldr (fn (x,y) => 1+y) 0 l;
fun addList l = foldr (op +) 0 l;
fun append(x,y) = foldr (op ::) y x;
fun smallest (first::rest) =
      foldr Int.min first rest;
fun split l =
      foldr (fn (first,(x,y)) => (first::y,x))
            ([],[]) l;

map kan definieras med hjälp av foldr:
fun map f l =
      foldr (fn (x,y) => f x::y) [] l;

Funktionen foldl
Det finns också en inbyggd funktion för iteration (svansrekursion) 
över listor:
(* foldl f e l
   TYPE: ('a*'b->'b) -> 'b -> 'a list -> 'b
   POST: f(xn,...f(x2,f(x1,e))...) om l=[x1,x2,...,xn]
*)
(* VARIANT: längden av l *)
fun foldl f e l =
  let
    fun foldlAux([],ack) = ack
      | foldlAux(first::rest,ack) =
          foldlAux(rest,f(first,ack))
  in
    foldlAux(l,e)
  end;

Exempel: fun rev l = foldl (op ::) [] l;

Funktionen filter
List.filter kan användas för att välja ut vissa delar av en lista. 
T.ex. List.filter (fn x => x>0) [1,~2,4,0] = [1,4]
(* filter p l
   TYPE: ('a -> bool) -> 'a list -> 'a list
   POST: En lista av de element i l som
         gör p sann *)
(* VARIANT: längden av l *)
fun filter p [] = []
  | filter p (first::rest) =
      if p first then
        first::filter p rest
      else
        filter p rest;
filter kan även definieras med hjälp av foldr:
fun filter p l =
 foldr (fn (x,y) => if p x then x::y else y) [] l

Högre ordningens funktioner för vektorer
I biblioteket Vector finns varianter av funktionerna map, foldr och
foldl som arbetar på vektorer i stället för listor.
Exempel: Summera heltal i en vektor (svansrekursivt):
(* sumVector a
   TYPE: int vector -> int
   POST: Summan av elementen i a *)
fun sumVector a = Vector.foldl (op +) 0 a;

Exempel: Bilda en lista av elementen i en vektor:
(* toList a
   TYPE: 'a vector -> 'a list
   POST: En lista av elementen i vektorn a. *)
fun toList a = Vector.foldr (op ::) [] a;

- toList #[18,12,~5,12,10];
> val it = [18, 12, ~5, 12, 10] : int list

